

QBITS GALAXY AD2375

There is Disquiet in the Galaxy as the Republic
First Order seeks to enforce new Taxes on
Interstellar Travel and Trade.



(M)arshal (G)eneral (C)aptain
Choose Skill Level and Side
(A)lliance or (R)epublic



Select Star System and <HyperSpace Jump> to Planet
Attack <?> Annex for UMD/Trade/Tech/Ores/Food/Fuel
Harvest Results and Reassign to Defend your
Star Sytems or use to Attack the Enemy

(G)ame Notes (M)enu

Press Any key



GALAXY AD2375



UMD **ALLIANCE**
:Land Based
Fuel:
Star Systems:

AD2375 Menu
(M)arshal (G)eneral (C)aptain
(N)ew (L)oad (S)ave (H)ighscore (E)xit
(A)lliance / (R)epublic

REPUBLIC UMD
Land Based:
Fuel:
Star Systems:

Commander: Captain

:Game Time

QBITS Introduction

In designing any form of game, the challenge is in getting the right balance between familiarity and ease of use while maintaining a level of difficulty with a variety of actions. Keeping the player engaged and enjoying the opportunities and satisfaction of completing a series of events.

QBITS GALXY AD2375

After choosing which side to play, the aim is to reduce your opponent's Star Systems to Zero. This begins by using the Left / Right Cursors to select an opponent's Star system from the Grid. Stars shown are circled either in Opponent or Player colours. Spacebar activates a Hyperspace jump to Orbit the major planet of the selected Star System.

Star Ship(s) Tactical information is displayed with Galaxy Grid reference on the left side of the screen, with Star System and Planet details on the right. Attack mode uses a random throw of the dice as shown in the circles placed below the areas showing Star and Planet Information

The outcome is a Victory and acquisition of the Star System or an abort due to insufficient WMD's to continue the attack. Victory secures an increase in Star Ships Tactical stock adding to or replacing lost items. The next move is to Transfer from Star Ship to Planet to raise output for a (H)arvest. Transfer from Planet to Star Ship is used to replenish and increase Tactical capability for future encounters.

QBITS AD2375 Tactical

There will be times when an Attack has reduced WMD's to critical levels. To replenish Stock and Weapons Select from the Grid a Star Systems with same colour as your own side. Hyperspace Jump to Orbit, as there is no need to Attack, use Star Ship Stock to increase the (H)arvest yields and build up your Star Ship Stocks. This can be repeated until ready to carry out further Attacks.


QBITS AD2375 Hazards

The first are views of a passing **Pulsar**, **Twin Star** formation or a **Nebula**, these create time delays limiting game time. Next is an encounter with a **Black Hole** which can incur loss of a Star Ship and crew plus twenty WMDs. **Encounter** is an attack by the Opponents Fighters. Each time this happens you lose a Fuel Cell to charge your Shields and Phasers. However, if you are successful in destroying an enemy fighter your Opponent loses the fighter plus 3 WMD and 1 Fuel Cell. **Time Slip** happens very rarely, but the Tactical Status of each side is swapped. You may have been winning or losing now the tables are turned.

QBITS AD2375 Design


Selection of Side (Alliance or Republic) and Skill level (Marshal or General or Captain) is made from the Main Menu. The side played by the current Player is held by variable Gp. The side chosen for the Gamer is held by variable Gp1 and Computer Opponent held by Gp2.

Simulation code generates the Opponents moves and is used to emulate the moves in Self Play Mode SPck=1 when activated by F1.



Identifying and providing convincing Computer Player moves is controlled by a number of randomised variables then restricting the generated result to within meaningful boundaries. See section **Opponent & Simulation Mode**, this generates inputs to the screen displays showing changes to the Star Ship and Star System Information tables as the Game progresses.

Star acquisition follows the pathways shown below. If selected Star System is Independent or on Opponent's side, you commence with an attack. If successful or Star System already acquired then Transfers can take place to replenish and/or acquire new Tactical Stock.



Encounter --- Raise Shields Charge Phasers (costs one fuel Cell)
Destroy the Enemy, Opponent lose fighter + 3 WMD's + 1 fuel Cell

QBITS AD2375 Control Keys

Moat of these are displayed as prompts relative to the actions taking place. Extensive use is made of the Cursor keys and Spacebar. For selection of Main Menu items and some other actions use bracketed single letters such as N/n for (N)ew L/l for (L)oad, H/h for **(H)**arvest etc.

QBITS AD2375 File Management

The set up for loading and running the AD2375 Prog is as one of a group of QBITS Progs selected from a QBITSProgs Menu. On start up the AD2375 Prog therefore expects to set a number of variables with information held by QBITConfig file. These are the gx,gy offsets to run under a QPCII environment a default to return to the QBITSProgs Menu and a number of other devices. The Opening lines of the AD2375 Program should be changed to reflect the user's personal setup.

Having set the screen and file management variables, both (L)oad and (S)ave will be able to offer Data storage to a range of Devices and Files 'AD2375Data_0 to 9'. The File Management carries out basic checks to validate device and filename returning 'DEVICE ERROR' or 'File NOT Found' and when Saving will prompt for 'Overwrite Y/N' if File is found to already exist. If no problems are found AD2375 Prog will Load from or Save to Selected Device and Filename.

1000 REMark AD2375_v3 [QBITS GALAXY AD2375 v3 2022]

1002 OPEN _IN#9,'ram2_QBITSConfig':INPUT#9,gx\gy\dn\$\\dev\$\\dn%\dm%
 1003 DIM Drv\$(15,5):FOR d=0 TO 15:INPUT#9,Drv\$(d):END FOR d:CLOSE#9


1005 **WHEN ERROR**
 1006 eck=1:CONTINUE
 1007 **END WHEN**

1009 DIM astro\$(70,10),astro(70,12) :REMark Star/Planet 1-70 Info
 1010 DIM side(2,6),min(6),max(6) :REMark Players Info : GTS h:m:s
 1011 DIM Gck\$(20),name\$(3,10),score(3,3) :REMark GTS h:m:s League Table

1013 wck=0:nck=1:pck=0:eck=0 :REMark WMD/New/Load/Save/ERROR checks
 1014 Gp=0:Gp1=1:Gp2=2:Gs=0:GSk=3 :REMark Player/Side/Skill
 1015 AWMD=0:ASSn=1:Alnt=12:APs=0 :REMark Alliance WMD/SShips/Fighters/Planets
 1016 RWMD=0:RSSn=1:RInt=12:RPs=0 :REMark Republic WMD/SShips/Fighters/Planets
 1017 dt=0:dw=3:ack=1:vck=0:t=1 :REMark Dicethrow:Del WMD:Attack/Transfer

1019 dP%=0:ws=1:Mes\$=' :REMark device file:Warp Speed:Messages
 1020 GTS=0:TD1=20:TD2=60:LTck=0 :REMark Game Time:Delays:League Table check

QBITS GALAXY AD2375 Screen Layouts



Warnings Shields & Phasers
Enemy Fighter Attacks

AD2375

Star Ship Info
Action

Dice Throw

Alliance Tactical
Capability

Skill Level

GALAXY GRID
x:219 y:44

Orbit

GALAXY AD2375 Simulator

STAR SYSTEM
Nashira

REPUBLIC
9.5b:Popu
3 :Moral
8 :Food
7 :Dres
8 :Tech
6 :Trade
3 :WMD
4 :Fuel

Nashira - Planetary Orbit

WMD [ALLIANCE]
41 :Land Based
17 : : 1
Fuel: : : 12
Star Systems: 9

Commander: General

Planet Info
Action

Dice Throw

Republic Tactical
Capability

Duration

'Yellow' Interactive Mode (F1) Simulation 'Red'

AD2375 Menu

(M)arshal (G)eneral (O)ptain
 (N)ew (L)oad (S)ave (H)ighscore (E)xit
 (A)lliance / (R)epublic

League Table

Skill	Time h:m:s	WMD	Stars	Comer
Marshal	01:24:32	1024	58	SPloy
Admiral	01:02:45	1185	62	CPloy
Captain	00:57:32	1296	66	QBITS

(H)ighscore

1022 REMark Part 1 SetUp - Intro - Menu

1024 MODE 4:Init_Win:GMenu

1026 DEFine PROCEDURE GMenu

1027 BLOCK#2,62,10,94,6,0:CLS#13:CLS#16

1028 INK#20,2:CURSOR#20,390,230:PRINT#20,' :Game Time'

1029 CSIZE#13,2,0:INK#13,7:QBold 13,12,56,2,'AD2375 Menu':CSIZE#13,0,0

1030 INK#13,6:CURSOR#13,16,26:PRINT#13,'(N)ew (L)oad (S)ave (H)ighscore (E)xit'

1031 ch=13:GChoice:SPck=0:GIcon:Alliance:Republic

1032 REPEAT lp

1033 k=CODE(INKEY\$(-1)):ch=13

1034 SELECT ON k

1035 =77,109:TD2=20 :GChoice :REMark Marshal

1036 =71,103:TD2=40 :GChoice :REMark General

1037 =67,99:TD2=60 :GChoice :REMark Captain

1038 =65,97:Gp1=1:Gp2=2 :GChoice:Init_Stats :REMark Alliance

1039 =82,114:Gp1=2:Gp2=1 :GChoice:Init_Stats :REMark Republic

1040 =78,110:mch=1:nck=0 :Init_New:GAD2375 :REMark (N)ew

1041 =76,108: :GLoad :REMark (L)oad

1042 =83,115: IF nck=0 :GSave :REMark (S)ave

1043 =72,104:LScore:PAUSE :GMenu :REMark (H)ighscore

1044 =68 :LTDefault:LSave :LScore:PAUSE:GMenu :REMark (D)efault Highscore

1045 = 100:LTRreset :LSave :LScore:PAUSE:GMenu :REMark (d) Reset highscore

1046 =69,101 :GExit :REMark (E)xit

1047 =32 : IF nck=0 :GMapInit:GAD2375 :REMark Continue Game

1048 END SELECT

1049 END REPEAT lp

1050 END DEFine

1052 DEFine PROCEDURE QBold(ch,cs,cx,cy,str\$)

1053 OVER#ch,1:sl=LEN(str\$)

1054 FOR c=1 TO sl

1055 FOR b=0 TO 1:CURSOR#ch,cx+b+c*cs,cy:PRINT#ch,str\$(c)

1056 END FOR c:OVER#ch,0

1057 END DEFine

1059 DEFine PROCEDURE GChoice

1060 IF TD2=20:Tm\$='Marshal':GSk=1:xsk= 38:sk\$='(M)':mx=4:my=2

1061 IF TD2=40:Tm\$='General':GSk=2:xsk=104:sk\$='(G)':mx=6:my=3

1062 IF TD2=60:Tm\$='Captain':GSk=3:xsk=170:sk\$='(C)':mx=8:my=4

1063 IF Gp1=1:xsd=62:sd\$='(A)':ELSE Gp1=2:xsd=140:sd\$='(R)'

1064 INK#20,2:CURSOR#20,8,230:PRINT#20,'Commander: ',Tm\$

1065 INK#ch,5:CURSOR#ch,62,38:PRINT#ch,'(A)lliance / (R)epublic'

1066 CURSOR#ch,38,14:PRINT#ch,'(M)arshal (G)eneral (C)aptain':INK#ch,7

1067 CURSOR#ch,cx,14:PRINT#ch,sk\$:CURSOR#ch,xsd,38:PRINT#ch,sd\$


1068 IF nck=0:Write 'To Continue Press Spacebar'

1069 END DEFine

1071 DEFine PROCEDURE GExit

1072 Write 'Exit Y/N':PAUSE:IF KEYROW(5)=64:LRUN dn\$:ELSE Write "

1073 END DEFine



1075 REMark Part 2 Game AD2370

1077 DEFine PROCEDURE GAD2375

```

1078 ch=1:CLS#ch:INK#ch,7:FOR i=1 TO 120:POINT#ch,RND(10 TO 200),RND( 5 TO 95)
1079 Gclk=DATE:clk1=DATE+TD1:clk2=DATE+TD2:clk3=DATE:haz=0:haz2=0
1080 SStats:LMap:n=RND(20 TO 50):bx=-9:by=-9:PMAP
1081 REPEAT lp
1082 GTime:StarMove:BEEP:k=CODE(INKEY$(10))
1083 Write 'Use ← → to Select Star System':BLOCK#16,16,3,34,4,7
1084 IF DATE>clk1:Hazards :clk1=DATE+TD1
1085 IF DATE>clk2:Gp=Gp2 :GSide:clk2=DATE+TD2 :REMark Opponent
1086 IF DATE>clk3 AND SPck=1:Gp=Gp1:GSide:clk3=DATE+TD2 :REMark Simulator
1087 SELECT ON k
1088 =32 :IF SPck=0 :Gp=Gp1:GSide :REMark [SBar] Select Star
1089 =71,103: :GNotes :REMark (G)ame Notes
1090 =77,109:GTS=(DATE-Gclk+GTS) :GMenu :REMark Skill/side/NEW/Load/Save
1091 =192:IF SPck=0 AND n>1 :n=n-1 :Gp=Gp1:SSInfo :REMark ← Star Left
1092 =200:IF SPck=0 AND n<70:n=n+1 :Gp=Gp1:SSInfo :REMark → Star Right
1093 =232:IF SPck=0:SPck=1 :GIcon:ELSE SPck=0:GIcon :REMark F1 Simulator
1094 =236:sn=n :GameTest :REMark F2 Graphics Test
1095 =240 :Blackhole:TimeSlip :REMark F3 cheat-Test
1096 =244 :Encounter:CLS#16 :REMark F4 Cheat-Test
1097 =248:side(Gp1,5)=side(Gp1,5)+10:SStats :REMark F5 Cheat-Tec/MODs
1098 END SElect
1099 END REPEAT lp
1100 END DEFine

```

1102 DEFine PROCEDURE GTime

```

1103 Gclk$=DATE$(DATE-Gclk+GTS):INK#20,2
1104 CURSOR#20,390,230:PRINT#20,Gclk$(13 TO 20)
1105 END DEFine

```

#16

1107 DEFine PROCEDURE Write(str\$)

```

1108 PRINT#16,FILL$(' ',16-(LEN(str$) DIV 2))&str$
1109 END DEFine

```


1111 DEFine PROCEDURE GMapInit

```

1112 ch=13:CLS#ch:INK#ch,0,2,3:SCALE#ch,100,0,0
1113 FOR h=12 TO 96 STEP 18:LINE#ch,6,h TO 310,h
1114 FOR v=16 TO 330 STEP 20:LINE#ch,v,3 TO v,98
1115 INK#ch,2:FOR i=1 TO 60 :POINT#ch,RND(60 TO 260),RND(30 TO 70)
1116 INK#ch,4:FOR i=1 TO 30 :POINT#ch,RND(80 TO 240),RND(40 TO 60)
1117 INK#ch,7:FOR i=1 TO 180:POINT#ch,RND(10 TO 300),RND(10 TO 90)
1118 END DEFine

```

#13



Note: Mode of Play Interactive or Simulation is set by Sim Player check 'SPck' and 'Gp' set to be Player or Opponent.

1120 DEFine PROCEDURE GSide

```

1121 WMD=side(Gp,5):ack=1:t=1:vck=1:Mes$=' ':INK#2,7:CURSOR#2,348,6
1122 IF SPck=1:PRINT#2,'Simulator' :GStarSel:GSwap:RETurn
1123 IF Gp=Gp2:PRINT#2,'Opponent':GStarSel:GSwap:RETurn
1124 IF Gp=Gp1:PRINT#2,'Gamer' :HJump:StarView:EOrbit:GPlay:GSwap
1125 END DEFine

```

1127 DEFine PROCEDURE Hazards

```

1128 haz=RND(1 TO 6):IF haz=haz2 :RETurn :ELSE haz2=haz
1129 IF haz=1 :BEEP 0, 3, 6,6000,5,0,0,0 :Pulsar :PAUSE 20
1130 IF haz=2 :BEEP 0, 9,18,3000,5,0,0,15:Binary :PAUSE 20
1131 IF haz=3 :Encounter
1132 IF haz=4 :BEEP 0,30,33,6000,5,2,5,15:Nebula :PAUSE 20
1133 IF haz=5 :BEEP 0,12,24,6000,8,6,8,15:Blackhole:PAUSE 20
1134 IF haz=6 AND DATE>clk2
1135 IF Gp1=1 AND APs<RPs+20:RETurn
1136 IF Gp1=2 AND RPs<APs+20:RETurn
1137 BEEP 0,12,32,3000,1,0,11,0:TimeSlip:BEEP 0,1,19,30,8,4,11,0
1138 PAUSE 20:clk2=DATE+TD2
1139 END IF
1140 END DEFine

```

1142 DEFine PROCEDURE Pulsar

```

1143 LOCal px,py,pr:pr=.3:ws=1:Write 'Passing Pulsar'
1144 FOR i=1 TO 32
1145 px=90+i*3:py=90-i*2:pr=i/6:GTime:StarMove
1146 INK 241:FILL 1:CIRCLE px,py,pr :FILL 0:PAUSE 3
1147 INK 7:FILL 1:CIRCLE px,py,pr*1.1:FILL 0:PAUSE 2
1148 CIRCLE px,py,pr*2,.3,PI/2:PAUSE 2
1149 INK 0:FILL 1:CIRCLE px,py,pr*2 :FILL 0
1150 END FOR i
1151 END DEFine

```



1153 DEFine PROCEDURE Binary

```

1154 LOCal a1,a2,bx,by,rx,ry:ws=1:Write 'Passing Binary Star System'
1155 a1=180:a2=0:rx=12:ry=4
1156 FOR i=1 TO 40
1157 bx=72-i*2:by=50+i/2
1158 x1=bx-rx*SIN(RAD(a1)):y1=by-ry*COS(RAD(a1))
1159 x2=bx-rx*SIN(RAD(a2)):y2=by-ry*COS(RAD(a2))
1160 sx=x1:sy=y1:sr=i/12:Star:a1=a1+6:IF a1>360:a1=0
1161 sx=x2:sy=y2:sr=i/6 :Star:a2=a2+6:IF a2>360:a2=0
1162 PAUSE 2+i/4:INK 0:FILL 1:CIRCLE x1,y1,i/4:FILL 0
1163 FILL 1:CIRCLE x2,y2,i/2:FILL 0:GTime:StarMove
1164 END FOR i
1165 END DEFine

```




1167 DEFine PROCEDURE Nebula

```

1168 LOCal x1,y1,x2,y2,x3,y3,x4,y4:ws=1:Write 'Passing Nebula'
1169 FOR a=1 TO 40
1170 x1=92-a*3:y1=50-a*1.5:x2=90-a*4:y2=48-a*1.1
1171 x3=94-a*3:y3=48-a*1.5:x4=92-a*3:y4=50-a*1.5:OVER -1
1172 INK 0,7,0:FILL 1:CIRCLE x1,y1,a*1.5,.3,PI/2:FILL 0
1173 INK 0,4,3:FILL 1:CIRCLE x2,y2,a,.3,PI/2:FILL 0
1174 INK 0,4,3:FILL 1:CIRCLE x3,y3,a*.7,.4,PI/2:FILL 0
1175 INK 0,2,0:FILL 1:CIRCLE x4,y4,a*.8,.4,-PI/4:FILL 0:OVER 0
1176 PAUSE 4+a/10:INK 0:FILL 1:CIRCLE x1,y1,a*.4,.3,PI/2:FILL 0
1177 GTime:StarMove:INK 7:FOR i=1 TO 5:POINT RND(30 TO 80),RND(5 TO 40)
1178 END FOR a
1179 END DEFine

```



```

1181 DEFine PROCEDURE Blackhole
1182 x=170:y=92:ws=1:Write 'WARNING! >>> Blackhole <<<'
1183 FOR a=8 TO 20:BHole:x=x-a/3:y=y-a/5:GTime:StarMove:PAUSE 5
1184 IF side(Gp1,5)>40
1185   side(Gp1,5)=side(Gp1,5)-20:Write 'One Ship and Crew Lost'
1186 ELSE
1187   Write 'Taking Evasive Action'
1188 END IF
1189 FOR a=20 TO 15 STEP -1:BHole:x=x+a:y=y-a/2:GTime:StarMove:PAUSE 2
1190 StarMove:SStats
1191 END DEFine

```

```

1193 DEFine PROCEDURE BHole
1194 OVER -1:INK 2:CIRCLE x,y,a*.3,.8,PI/2
1195 ARC x,y+a*.2 TO x-a*.3,y,PI/2 TO x,y-a*.3,PI/2 TO x+a*.4,y,PI/2
1196 ARC TO x,y+a*.4,PI/2 TO x-a*.6,y-a*.6,PI/2
1197 ARC x,y+a*.2 TO x-a*.4,y,PI/2 TO x,y-a*.4,PI/2 TO x+a*.5,y,PI/2
1198 ARC TO x,y+a*.5,PI/2 TO x-a*.7,y-a*.7,PI/2
1199 OVER 0:PAUSE 5:INK 0:FILL 1:CIRCLE x,y,a*4:FILL 0
1200 END DEFine


```



```

1202 DEFine PROCEDURE TimeSlip
1203 Write 'Time Slip':BLOCK#2,60,10,96,6,0:INK 7
1204 FOR a=2 TO 8 STEP 2:CIRCLE 96,50,a*.4,.2,PI/2:PAUSE 3
1205 FOR n=1 TO 70
1206   IF astro(n,7)=1:astro(n,7)=2:ELSE IF astro(n,7)=2:astro(n,7)=1
1207 END FOR n
1208 FOR i=1 TO 5
1209   a=side(1,i):side(1,i)=side(2,i):side(2,i)=a
1210 END FOR i
1211 FOR b=16 TO 2 STEP -1
1212   LINE 96-b*3,50 TO 96+b*3,50:LINE 96-b*5,50+b*3 TO 96+b*5,50+b*3
1213   LINE 96-b*5,50-b*3 TO 96+b*5,50-b*3:PAUSE 5
1214 END FOR b
1215 FOR a=1 TO 2:side(a,7)=10:IF side(a,5)<20:side(a,5)=20:END IF
1216 GTime:GMapInit:LMap:bx=-9:by=-9:PMap:SStats
1217 END DEFine

```




```

1219 DEFine PROCEDURE LMap
1220 LOCAL n,nx,ny:ch=13
1221 FOR n=1 TO 70
1222   nx=astro(n,8):ny=astro(n,9):INK#ch,7
1223   IF astro(n,7)=1:CIRCLE#ch,nx,ny,3:INK#ch,5:CIRCLE#ch,nx,ny,6
1224   IF astro(n,7)=2:CIRCLE#ch,nx,ny,3:INK#ch,3:CIRCLE#ch,nx,ny,6
1225 END FOR n
1226 END DEFine

```

Star List



```

1228 DEFine PROCEDURE PMap
1229 ch=13:INK#ch,7:OVER#ch,-1 :REMark bx by Base Coordinates
1230 LINE#ch,bx,5 TO bx,95:LINE#ch,12,by TO 306,by
1231 CIRCLE#ch,bx,by,3:INK#ch,7:CIRCLE#ch,bx,by,9:OVER#ch,0
1232 END DEFine

```

Star Position Crosswires




```

1234 DEFINE PROCEDURE Encounter
1235 BEEP 0,1,9,100,-5,8,0,0:PMap:bx=-9:by=-9:PMap
1236 Write 'Enemy Fighters Approaching'
1237 ch=2:BLOCK#ch,62,18,94,0,0:BLOCK#ch,62,18,346,0,0
1238 FOR alm=1 TO 12
1239   col=2:IF alm MOD 2=0:col=0
1240   EAS col,41,96:EAS col,123,96:PAUSE 5
1241 END FOR
1242 BEEP 0,60,3,900,1,0,0,0:INK 7:DPS 11:DPS 12
1243 Write 'Abort<Esc>:Target ← → Fire':BLOCK#16,16,3,130,4,7
1244 IF side(Gp1,6)>5:side(Gp1,6)=side(Gp1,6)-2:ELSE side(Gp1,6)=3
1245 RCharge:cs=0:cp=0:SStats:BEEP:turns=RND(3 TO 5):des=0
1246 FOR att=1 TO turns
1247   fx=RND(1 TO 3):IF Gp1=1:CG=fx:ELSE CG=fx+3
1248   xx=fx*60-24+8*RND(-1 TO 1) :yy=95:tx=95
1249   REPEAT act_lp
1250     IF side(1,5)<9 OR side(2,5)<9 OR cs>64 OR cp>64 :EXIT att
1251     IF side(1,6)<2 OR side(2,6)<2 OR Alnt<1 OR RInt<1:EXIT att
1252     IF xx<10 OR xx>190 OR yy<5 OR des=1:des=0 :EXIT att_lp
1253     IF fx=1:xx=xx+RND(-3 TO 12)
1254     IF fx=2:xx=xx+RND(-12 TO 12)
1255     IF fx=3:xx=xx+RND(-12 TO 3)
1256     IF SPck=1 AND xx>90 AND xx<100:PFire tx,xx,yy
1257     yy=yy-RND(2 TO 5)
1258     INK 7:Enemy CG:INK 4:PSights tx,yy
1259     k=CODE(INKEY$(5)) :IF yy>30 AND RND(2)=1:FFire xx,yy
1260     INK 0:Enemy CG:INK 0:PSights tx,yy
1261     SELECT ON k
1262       = 27:EXIT att
1263       = 32:PFire tx,xx,yy:INK 7
1264       =192:tx=tx-5:IF tx< 20:tx= 20
1265       =200:tx=tx+5:IF tx>160:tx=160
1266     END SELECT
1267   END REPEAT act_lp
1268 END FOR att
1269 BLOCK#2,62,18,94,0,0:BLOCK#2,62,18,344,0,0
1270 sn=RND(1 TO 70):Gp=Gp1:HJump:DPS 11:DPS 12
1271 END DEFINE

```



```

1273 DEFINE PROCEDURE EAS(col,ax,ay)
1274 INK#ch,col:FILL#ch,1:CIRCLE#ch,ax,ay,2:FILL#ch,0
1275 FOR a=0 TO .25 STEP .25
1276   FOR b=1 TO 3:ARC#ch,ax-b*2-a,ay+3 TO ax-b*2-a,ay-3,PI
1277   FOR b=1 TO 3:ARC#ch,ax+b*2+a,ay+3 TO ax+b*2+a,ay-3,-PI
1278 END FOR a
1279 END DEFINE


```



```

1281 DEFINE PROCEDURE RCharge
1282 FOR i=1 TO 8
1283   BLOCK#11,i*8,4,20,90,4 :BLOCK#12,i*8,4,70-i*8,90,4:PAUSE 5
1284   BLOCK#11,i*8,60,72-i*8,7:BLOCK#12,i*8,18,72-i*8,7:PAUSE 5
1285 END FOR i
1286 END DEFINE



```



```

1288 DEFine PROCEDURE DPS(ch)
1289 IF ch=11:CLS#11:INK#11,7:ch1=14:RESTORE 1302:fx1=36:fx2=48:fx4=8
1290 IF ch=12:CLS#12:INK#12,7:ch1=15:RESTORE 1304:fx1=12:fx2=36:fx4=70
1291 CLS#ch1:FOR i=1 TO 5:BLOCK#ch1,5,14,8+i*10,2,4
1292     FOR i=1 TO 6:BLOCK#ch1,3,10,4+i*10,4,4
1293 FOR i=1 TO 2
1294     READ x1,y1,x2,y2,x3,y3,x4,y4,x5,y5,x6,y6,x7,y7,x8,y8
1295     LINE#ch,x1,y1 TO x2,y2 TO x3,y3:ARC#ch TO x4,y4,PI/2
1296     LINE#ch TO x5,y5 TO x6,y6 TO x7,y7:ARC#ch TO x8,y8,-PI/2:LINE#ch TO x1,y1
1297 END FOR i
1298 FOR i=1 TO 8:BLOCK#ch,8,2,fx4+2,6+i*8,2
1299 FOR i=1 TO 7:BLOCK#ch,12,4,fx4,9+i*8,2
1300 INK#ch,4:C ursor#ch,fx1,80:PRINT#ch,'PHASERS':str$='SHIELDS'
1301 INK#ch,6:FOR i=1 TO 7:CURSOR#ch,fx2,i*9:PRINT#ch,str$(i)
1302 DATA 7,22,13,22,13,16,16,14,64,14,64,2,17,2,7,12 :REMark Phaser #11
1303 DATA 32,22,64,22,64,90,54,100,7,100,7,90,29,90,32,87 :REMark Sheild
1304 DATA 2,2,2,14,51,14,54,17,54,22,60,22,60,12,50,2 :REMark Phasor #12
1305 DATA 60,100,60,90,38,90,35,87,35,22,2,22,2,90,12,100 :REMark Shield
1306 END DEFine

```



```

1308 DEFine PROCEDURE FFire(xx,yy)
1309 BEEP 3000,1,19,69,81,4,2,0:INK 2:cs=cs+2:tm=RND(-15 TO 15)
1310 BLOCK#11,8,cs,60,8,0:BLOCK#12,8,cs,18,8,0
1311 Phaser 2,xx,yy-2,95+tm,2:PAUSE 2:Phaser 0,xx,yy-2,95+tm,2
1312 END DEFine

```

```

1314 DEFine PROCEDURE PSights(tx,yy)
1315 LINE tx-8,y+8 TO tx-3,y+3:LINE tx+3,y+3 TO tx+8,y+8
1316 LINE tx-8,y-8 TO tx-3,y-3:LINE tx+3,y-3 TO tx+8,y-8
1317 END DEFine

```

```



1319 DEFine PROCEDURE PFire(tx,xx,yy)
1320 BEEP 2000,1,255,200,4,2,0,0,0:INK 4:cp=cp+4
1321 BLOCK#11,cp,4,84-cp,90,0:BLOCK#12,cp,4,6,90,0
1322 Phaser 4,95,2,tx,yy:PAUSE 2:Phaser 0,95,2,tx,yy
1323 IF tx<xx-1 OR tx>xx+1:RETurn
1324 BEEP 20000,5,12,5,4,0,0,des=1 :REMark Enemy Destroyed
1325 INK 4:FOR i=3 TO 9 :CIRCLE xx,yy,i:PAUSE 2
1326 INK 0:PAUSE 5:FILL 1:CIRCLE xx,yy,15:FILL 0
1327 FOR i=1 TO 8:POINT RND(xx-9 TO xx+9),RND(yy-9 TO yy+9)
1328 IF Gp1=1:side(2,5)=side(2,5)-3:ELSE side(1,5)=side(1,5)-3
1329 IF Gp1=1:side(2,6)=side(2,6)-1:ELSE side(1,6)=side(1,6)-1
1330 IF Gp1=1:RInt=RInt-1:SStats :ELSE AInt=AInt-1:SStats
1331 END DEFine

```

```

1333 DEFine PROCEDURE Phaser(col,xx,yy,xt,yt)
1334 INK col:FILL 1:LINE xx,yy TO xt-.5,yt TO xt+.5,yt TO xx,yy:FILL 0
1335 END DEFine

```



1337 **DEfINE PROCEDURE Enemy(CG)**

1338 **SElECT ON CG**

1339 =1:RF1 1,xx,yy

1340 =2:RF2 1,xx,yy

1341 =3:RF3 1,xx,yy

1342 =4:AF1 1,xx,yy

1343 =5:AF2 1,xx,yy

1344 =6:AF3 1,xx,yy

1345 **END SElECT**

1346 **END DEfINE**

1348 **DEfINE PROCEDURE AF1(ch,x,y)**

1349 CIRCLE#ch,x,y+2,1.2,.6,PI:LINE#ch,x-.5,y+3 TO x-4,y+5

1350 FILL#ch,1:LINE#ch,x,y TO x+4,y-4.5 TO x+3,y-5 TO x-4,y-1 TO x,y:FILL#ch,0

1351 FILL#ch,1:CIRCLE#ch,x-4.5,y+3,3,.3,PI/3:FILL#ch,0

1352 FILL#ch,1:LINE#ch,x-5,y TO x-9,y-2 TO x-11,y TO x-9,y+3 TO x-5,y:FILL#ch,0

1353 CIRCLE#ch,x-4,y,1.2,.6,PI:LINE#ch,x+3,y TO x,y+2

1354 **END DEfINE**



1356 **DEfINE PROCEDURE AF2(ch,x,y)**

1357 FILL#ch,1:LINE#ch,x-2,y-1 TO x-1,y-4 TO x+1,y-4 TO x+2,y-1 TO x-2,y-1

1358 FILL#ch,0:LINE#ch,x-10,y-1.5 TO x-6,y-.5:CIRCLE#ch,x-4.5,y,1.6

1359 FILL#ch,1:CIRCLE#ch,x,y,2.5

1360 FILL#ch,0:LINE#ch,x+10,y-1.5 TO x+6,y-.5:CIRCLE#ch,x+4.5,y,1.6

1361 **END DEfINE**



1363 **DEfINE PROCEDURE AF3(ch,x,y)**

1364 CIRCLE#ch,x,y+2,1.2,.6,PI:LINE#ch,x+.5,y+3 TO x+4,y+5

1365 FILL#ch,1:LINE#ch,x,y TO x-4,y-4.5 TO x-3,y-5 TO x+4,y-1 TO x,y:FILL#ch,0

1366 FILL#ch,1:CIRCLE#ch,x+4.5,y+3,3,.3,-PI/3:FILL#ch,0

1367 FILL#ch,1:LINE#ch,x+5,y TO x+9,y-2 TO x+11,y TO x+9,y+3 TO x+5,y:FILL#ch,0

1368 CIRCLE#ch,x+4,y,1.2,.6,PI:LINE#ch,x-3,y TO x,y+2

1369 **END DEfINE**



1371 **DEfINE PROCEDURE RF1(ch,x,y)**

1372 LINE#ch,x+2,y-3 TO x+3.5,y-5 TO x+5,y+1 TO x+3,y+6.4 TO x+1,y+3

1373 FILL#ch,1:CIRCLE#ch,x,y,2.5:FILL#ch,0

1374 LINE#ch,x-6,y TO x-3,y+6 TO x-2,y TO x-3.5,y-6 TO x-6,y

1375 **END DEfINE**



1377 **DEfINE PROCEDURE RF2(ch,x,y)**

1378 LINE#ch,x-6,y TO x-4,y+5 TO x-4,y-5 TO x-6,y :REMark L Wing

1379 FILL#ch,1:CIRCLE#ch,x,y,2.5:FILL#ch,0 :REMark Sphere

1380 LINE#ch,x+6,y TO x+4,y+5 TO x+4,y-5 TO x+6,y :REMark R Wing

1381 **END DEfINE**



1383 **DEfINE PROCEDURE RF3(ch,x,y)**

1384 LINE#ch,x-2,y-3 TO x-3.5,y-5 TO x-5,y+1 TO x-3,y+6.4 TO x,y+1

1385 FILL#ch,1:CIRCLE#ch,x,y,2.5:FILL#ch,0

1386 LINE#ch,x+6,y TO x+3,y+6 TO x+2,y TO x+3.5,y-6 TO x+6,y

1387 **END DEfINE**



```

1389 DEFine PROCEDURE HJump
1390 Write 'Hyperspace Jump to '&astro$(sn):CURSOR 0,43
1391 GTime:wx=96:wy=50:ws=3:INK 241:BEEP 0,9,36,1000,1,0,5,5
1392 FOR a=42 TO 24 STEP -3:CIRCLE wx,wy,a:PAUSE a/4
1393 FOR a=1 TO 15
1394   FOR b=0 TO 3:CIRCLE wx,wy,20+b*15,.6,PI/2
1395   StarMove:BLOCK#2,60,10,96,6,0:INK 241:PAUSE 2
1396 END FOR a
1397 BEEP 0,60,90,3000,1,5,6,6 :FOR a=12 TO 1 STEP -1:ws=a:StarMove:PAUSE 14-a
1398 GTime:SSInfo:INK 7
1399 END DEFine

```



```

1401 DEFine PROCEDURE StarMove
1402 hoz=ws:ver=ws:INK#2,7
1403 CURSOR#2,96,6:PRINT#2,'Warp:':FILL#(' ',2-LEN(ws))&ws
1404 IF ws>0:INK#1,7:FOR w=1 TO ws*2:POINT#1,RND(40 TO 160),RND(30 TO 70)
1405 SCROLL#4,-ver*2:SCROLL#9,ver*2:PAN#6,-hoz*2:PAN#7,hoz*2
1406 PAN#3,-hoz*2:SCROLL#3,-ver:PAN#5,hoz*2:SCROLL#5,-ver
1407 PAN#8,-hoz*2:SCROLL#8, ver:PAN#10,hoz*2:SCROLL#10,ver
1408 END DEFine

```

```

1410 DEFine PROCEDURE StarView
1411 Write 'Approching '&astro$(sn):sy=80:sx=16+(astro(sn,11)*40)
1412 FOR r=1 TO 9:INK 0:FILL 1:CIRCLE sx,sy,r*2:FILL 0:sr=r:Star:StarMove:PAUSE 2
1413 END DEFine

```

```

1415 DEFine PROCEDURE Star
1416 INK 7:FILL 1:CIRCLE sx,sy,sr:FILL 0:INK 2,7,3:CIRCLE sx,sy,sr
1417 INK 2:FILL 1:CIRCLE sx-sr/2,sy-sr/3,sr/6,.3,RAD(45):FILL 0
1418 END DEFine

```



```

1420 DEFine PROCEDURE ASS(ch,col,x,y)
1421 INK#ch,col:FILL#ch,1:CIRCLE#ch,x,y,8,.5,PI/2:FILL#ch,0
1422 FILL#ch,1:CIRCLE#ch,x,y+4,4,.6,PI/2:FILL#ch,0
1423 CIRCLE#ch,x-9,y,3,.6,PI:CIRCLE#ch,x+9,y,3,.6,PI
1424 FILL#ch,1:CIRCLE#ch,x,y-5,6,.2,PI:FILL#ch,0
1425 INK#ch,0:LINE#ch,x-2.5,y+4 TO x+2.5,y+4:INK#ch,7
1426 END DEFine

```



```

1428 DEFine PROCEDURE RSS(ch,col,x,y)
1429 INK#ch,col:LINE#ch,x-9,y-2 TO x,y-7.5 TO x+9,y-2
1430 FILL#ch,1:LINE#ch,x-2.5,y+5 TO x+2.5,y+5 TO x+4,y+2.5 TO x+12,y-.5
1431 LINE#ch TO x,y-4 TO x-12,y-.5 TO x-4,y+2.5 TO x-2,y+5:FILL#ch,0
1432 FILL#ch,1:CIRCLE#ch,x,y+8,6,.2,PI/2:FILL#ch,0
1433 INK#ch,0:LINE#ch,x-2.8,y+2 TO x+3,y+2:INK#ch,7
1434 END DEFine

```



```

1436 DEFINE PROCEDURE EOrbit
1437 IF side(Gp,6)>3:side(Gp,6)=side(Gp,6)-RND(1 TO 3):ELSE side(Gp,6)=3
1438 SFuel:Write astro$(sn)&' - Planetary Orbit'
1439 x=96:INK#2,7:CURSOR#2,96,6:PRINT#2,'Impulse '
1440 FOR a=1 TO 18
1441 POINT RND(10 TO 180),RND(60 TO 80):SCROLL -1:GTime
1442 IF astro(sn,11)=1 :x=18+a*3
1443 IF astro(sn,11)=3 :x=182-(a*3)
1444 INK astro(sn,10) :y=24-a*3
1445 FILL 1:CIRCLE x,y,a*2:FILL 0:PAUSE a/3:INK 0
1446 FILL 1:CIRCLE x,y,a*2:FILL 0
1447 INK 7 :POINT RND(40 TO 160),RND(30 TO 40)
1448 END FOR a
1449 GTime:rx=astro(sn,10)-20
1450 IF rx<8 AND astro(n,11)<>3:c1=5:c3=2.5:ELSE c1=1:c3=1.2:INK#1,7
1451 IF rx>18 AND astro(n,11)=3:c2=6:ELSE c2=1
1452 FILL#1,1:CIRCLE#1,rx,15,c1 :FILL#1,0
1453 FILL#1,1:CIRCLE#1,170+rx,52,c2 :FILL#1,0
1454 FILL#1,1:CIRCLE#1,118+rx,110-rx,c3:FILL#1,0
1455 IF rx<6:INK#1,2:ARC#1,114+rx,112-rx TO 122+rx,108-rx,PI/2:INK#1,7
1456 FOR b=1 TO 14
1457 INK 7:POINT RND(10 TO 180),RND(10 TO 20)
1458 IF Gp=1:x=96-40:y=50:ELSE x=96+40:y=50
1459 INK astro(sn,10):FILL 1:CIRCLE 96,-200+b,198:FILL 0
1460 INK 241:CIRCLE x,y,b-2:LINE x-b,y TO x+b,y:LINE x,y-b TO x,y+b:PAUSE b/4
1461 INK 0:CIRCLE x,y,b-2:LINE x-b,y TO x+b,y:LINE x,y-b TO x,y+b
1462 END FOR b
1463 IF Gp=1:ASS 1,7,56,52:ELSE RSS 1,7,136,52
1464 INK#2,7:CURSOR#2,96,6:PRINT#2,'Orbit ':BEEP
1465 END DEFINE

```



```

1467 DEFINE PROCEDURE SFuel
1468 BLOCK#17,60,5,32,45,0:FOR i=1 TO side(1,6):BLOCK#17,3,5,26+i*6,45,5
1469 BLOCK#18,60,5, 6,45,0:FOR i=1 TO side(2,6):BLOCK#18,3,5,68-i*6,45,3
1470 END DEFINE

```



```

1472 DEFINE PROCEDURE LOrbit
1473 Write 'Leaving Orbit':INK#2,7:CURSOR#2,96,6:PRINT#2,'Impulse '
1474 BEEP 0,60,18,900,1,0,0,0:IF Gp=1:ASS 1,0,56,52:ELSE RSS 1,0,136,52
1475 IF Gp=1:x=60+42-(b*3):y=50:ELSE x=140-42+(b*3):y=55
1476 FOR b=15 TO 1 STEP -2
1477 INK 241:CIRCLE x,y,b-4:LINE x-b,y TO x+b,y:LINE x,y-b TO x,y+b:PAUSE a/2
1478 INK 0:CIRCLE x,y,b-4:LINE x-b,y TO x+b,y:LINE x,y-b TO x,y+b:INK 7
1479 CURSOR 0,80:SCROLL -1,1:PAN#6, -1
1480 CURSOR 0,40:SCROLL 2,2:PAN#7, 1
1481 FOR i=1 TO 3:POINT RND(20 TO 180),RND(10 TO 40)
1482 END FOR b
1483 GTime:CLS#16:DiceCLS:ws=1:FOR a=1 TO 4:StarMove:PAUSE 2
1484 END DEFINE

```



```

1486 DEFine PROCEDURE GPlay
1487 REPeat act_ip
1488 IF APs=0 OR RPs=0:EXIT act_ip
1489 IF ack=1 AND astro(n,7)=Gp:ack=0:Mes$="%4i %4Transfer%2":SSInfo
1490 IF ack=1 AND astro(n,5)=0 :ack=0:Mes$="%4i %4Transfer%2":Victory
1491 IF ack=1 AND side(Gp,5)>dw+3
1492 Mes$=" Attack Mode ":SSInfo:RollDice:SSInfo
1493 END IF
1494 IF ack=1 AND side(Gp,5)<dw+3
1495 ack=0:Write 'Abort low on WMDs':PAUSE 50:EXIT act_ip
1496 END IF
1497 IF ack=0:ch=11:BLOCK#ch,10,60,74,30,0:CUSOR#ch,74,20+t*10:PRINT#ch,'>'
1498 IF ack=0:ch=12:BLOCK#ch,10,60, 4,30,0:CUSOR#ch, 4,20+t*10:PRINT#ch,'<'
1499 GTime:k=CODE(INKEY$(25))
1500 SElect ON k
1501 =192:TranPS :REMark Transfer Planet > Star
1502 =200:TranSP :REMark Transfer Star > Planet
1503 =208:t=t-1:IF t<1:t=6 :REMark Select Food – Fuel etc
1504 =216:t=t+1:IF t>6:t=1
1505 = 32:EXIT act_ip
1506 = 78,104:IF ack=0 AND vck=1:Harvest :REMark (H)arvest
1507 END SElect
1508 END REPeat act_ip
1509 END DEFine

```

```

1511 DEFine PROCEDURE TranSP
1512 min(5)=16:min(6)=5:FOR i=1 TO 4:min(i)=1+side(Gp,5) DIV 20
1513 max(1)=20:max(2)=8:max(3)=12:max(4)=16:max(5)=16:max(6)=5
1514 IF side(Gp,t)<=min(t) OR astro(n,t)=max(t):RETurn
1515 side(Gp,t)=side(Gp,t)-1:astro(n,t)=astro(n,t)+1:SSInfo
1516 END DEFine

```

```

1518 DEFine PROCEDURE Harvest
1519 ch=12:vck=0:PAUSE 20:BLOCK#ch,90,10,0,90,0:Popu=1+INT(astro(n,12))
1520 astro(n,1)=astro(n,1)+Popu DIV 2 :IF astro(n,1)>20:astro(n,1)=20
1521 astro(n,2)=astro(n,2)+Popu DIV 2 :IF astro(n,2)> 8:astro(n,2)= 8
1522 astro(n,3)=astro(n,3)+Popu DIV 3 :IF astro(n,3)>12:astro(n,3)=12
1523 astro(n,4)=astro(n,4)+Popu DIV 3 :IF astro(n,4)>16:astro(n,4)=16
1524 astro(n,5)=astro(n,5)+astro(n,3) :IF astro(n,5)>16:astro(n,5)=16
1525 astro(n,6)=astro(n,6)+astro(n,3) DIV 3:IF astro(n,6)> 5:astro(n,6)= 5
1526 astro(n,12)=astro(n,1)/2.1:SSInfo
1527 Write 'To leave Orbit press ':BLOCK#16,16,3,150,4,7:PAUSE 30
1528 END DEFine

```

```

1530 DEFine PROCEDURE TranPS
1531 min(6)=0:FOR i=1 TO 5:min(i)=1
1532 max(1)=20:max(2)=8:max(3)=12:max(4)=16:max(5)=99:max(6)=10
1533 IF astro(n,t)=min(t) OR side(Gp,t)=max(t):RETurn
1534 side(Gp,t)=side(Gp,t)+1:astro(n,t)=astro(n,t)-1
1535 IF astro(n,1)<astro(n,12):astro(n,12)=astro(n,1):PTMoral
1536 Ship=1+(side(Gp,5)DIV 20):SSInfo
1537 END DEFine

```

1539 DEFine PROCEDURE SStats

```

1540 AWMD=0:APs=0:ASSn=0:RWMD=0:RPs=0:RSSn=0:SFuel
1541 FOR i=1 TO 70:IF astro(i,7)=1:APs=APs+1:AWMD=AWMD+astro(i,5)
1542 FOR i=1 TO 70:IF astro(i,7)=2:RPs=RPs+1:RWMD=RWMD+astro(i,5)
1543 IF wck=0:Set_WMD:wck=1
1544 BLOCK#17,20,10,0,12,0:CUSOR#17,0,12:PRINT#17,AWMD :REMark All Land WMD
1545 BLOCK#17,20,10,0,22,0:CUSOR#17,0,22:PRINT#17,side(1,5) :REMark SS WMD
1546 ASSn=1+side(1,5) DIV 21:CUSOR#17,78,22:PRINT#17,' ':ASSn :REMark All SShips
1547 CUSOR#17,78,32:PRINT#17,FILL$(' ',2-LEN(Alnt))&Alnt :REMark All Fighters
1548 CUSOR#17,78,52:PRINT#17,FILL$(' ',2-LEN(APs))&APs :REMark All SSsystems
1549 CUSOR#18,78,12:PRINT#18,FILL$(' ',3-LEN(RWMD))&RWMD :REMark Rep Land WMD
1550 CUSOR#18,84,22:PRINT#18,FILL$(' ',2-LEN(side(2,5)))&side(2,5) :REMark SS WMD
1551 RSSn=1+side(2,5) DIV 21:CUSOR#18,12,22:PRINT#18,RSSn :REMark Rep SShip
1552 CUSOR#18,6,32:PRINT#18,FILL$(' ',2-LEN(RInt))&RInt :REMark Rep Fighters
1553 CUSOR#18,6,52:PRINT#18,FILL$(' ',2-LEN(RPs))&RPs :REMark Rep SSsystems
1554 END DEFine

```



1556 DEFine PROCEDURE Set_WMD

```

1557 IF AWMD>RWMD:side(2,5)=20+AWMD-RWMD:side(1,5)=20
1558 IF RWMD>AWMD:side(1,5)=20+RWMD-AWMD:side(2,5)=20
1559 IF RWMD=AWMD:side(1,5)=20:side(2,5)=20
1560 side(1,6)=10:side(2,6)=10:SFuel
1561 END DEFine

```

1563 DEFine PROCEDURE RollDice

```

1564 ch=2:GTime:IF SPck=0 AND Gp=Gp1:Write 'Press any key to Fire':PAUSE
1565 xt=RND(80 TO 120):ys=46:IF Gp=1:xs=60:ELSE xs=132
1566 FOR a=1 TO RND(4 TO 6)
1567 dx= 5:DiceA:PAUSE 4:DiceB:Phaser 2,xs,ys,xt,12:BEEP 0,1,19,69,81,43,29,0
1568 dx=159:DiceA:PAUSE 4:DiceB:Phaser 0,xs,ys,xt,12:BEEP
1569 END FOR a
1570 OVER#ch,1:dt1=RND(1 TO 8):dx=5:DiceA:CUSOR#ch,dx,dy,-3,-4:PRINT#ch,dt1
1571 dt2=RND(1 TO 8):dx=159:DiceA:CUSOR#ch,dx,dy,-3,-4:PRINT#ch,dt2:OVER#ch,0
1572 IF dt1>dt2:astro(n,5)=astro(n,5)-dw:IF astro(n,5)<1:astro(n,5)=0
1573 IF dt1=dt2 OR dt1<dt2:side(Gp,5)=side(Gp,5)-dw
1574 END DEFine

```

1576 DEFine PROCEDURE DiceA

```

1577 dy=39:l=2.5:INK#ch,0:FILL#ch,1:CIRCLE#ch,dx,dy,3.5:FILL#ch,0:INK#ch,7
1578 LINE#ch,dx-l,dy+l TO dx+l,dy+l TO dx+l,dy-l TO dx-l,dy-l TO dx-l,dy+l
1579 END DEFine

```



1581 DEFine PROCEDURE DiceB

```

1582 dy=39:l=2.5:INK#ch,0:FILL#ch,1:CIRCLE#ch,dx,dy,3.5:FILL#ch,0:INK#ch,7
1583 LINE#ch,dx,dy+l TO dx+l,dy TO dx,dy-l TO dx-l,dy TO dx,dy+l
1584 END DEFine

```



1586 DEFine PROCEDURE DiceCLS

```

1587 ch=2:INK#ch,0:dy=39:dx=5:FILL#ch,1:CIRCLE#ch,dx,dy,4:FILL#ch,0
1588 dx=159:FILL#ch,1:CIRCLE#ch,dx,dy,4:FILL#ch,0:INK#ch,7
1589 END DEFine

```



```

1591 DEFine PROCEDURE SSInfo
1592 SStats:PTInfo:ch=11:IF side(Gp,6)<3:side(Gp,6)=3
1593 IF Gp=1:Gp$='ALLIANCE':ss=RSSn:INK#ch,5:INK#2,5
1594 IF Gp=2:Gp$='REPUBLIC':ss=FSSn:INK#ch,3:INK#2,3
1595 CLS#ch:QBold ch,7,0,0,Gp$:SFuel:CIRCLE#2,5,39,4:SSMoral
1596 PRINT#ch,' SShip:':FILL$(' ',4-LEN(ss))&ss
1597 PRINT#ch,' Moral:':FILL$(' ',4-LEN(smv))&smv
1598 PRINT#ch,' Food:':FILL$(' ',4-LEN(side(Gp,1)))&side(Gp,1)
1599 PRINT#ch,' Ores:':FILL$(' ',4-LEN(side(Gp,2)))&side(Gp,2)
1600 PRINT#ch,' Tech:':FILL$(' ',4-LEN(side(Gp,3)))&side(Gp,3)
1601 PRINT#ch,' Trade:':FILL$(' ',4-LEN(side(Gp,4)))&side(Gp,4)
1602 PRINT#ch,' WMD:':FILL$(' ',4-LEN(side(Gp,5)))&side(Gp,5)
1603 PRINT#ch,' Fuel:':FILL$(' ',4-LEN(side(Gp,6)))&side(Gp,6)
1604 CURSOR#ch,4,90:PRINT#ch,Mes$
1605 IF ack=0 AND vck=1:ch=12:CURSOR#ch,0,90:PRINT#ch,' (H)arvest '
1606 END DEFINE

```

```

1608 DEFine PROCEDURE SSMoral
1609 smv=1+INT((side(Gp,1)+side(Gp,2)+side(Gp,3))/DIV 7):REMark Ship Moral value
1610 END DEFINE

```

```

1612 DEFine PROCEDURE Victory
1613 Write "Transfer then - (H)arvest":BEEP
1614 side(Gp,1)=side(Gp,1)+astro(n,1)/DIV 3 :IF side(Gp,1)>20:side(Gp,1)=20
1615 side(Gp,2)=side(Gp,2)+astro(n,2)/DIV 4 :IF side(Gp,2)> 8:side(Gp,2)= 8
1616 side(Gp,3)=side(Gp,3)+astro(n,3)/DIV 4 :IF side(Gp,3)>12:side(Gp,3)=12
1617 side(Gp,4)=side(Gp,4)+astro(n,4)/DIV 4 :IF side(Gp,4)>16:side(Gp,4)=16
1618 side(Gp,5)=side(Gp,5)+1+INT(astro(n,12)):IF side(Gp,5)>99:side(Gp,5)=99
1619 astro(n,7)=Gp:LMap:vck=1:SSInfo
1620 END DEFINE

```

```

1622 DEFine PROCEDURE PTMoral
1623 pmv=1+INT((astro(n,1)+astro(n,2)+astro(n,3)+astro(n,5))/DIV 9)
1624 END DEFINE

```

```

1626 DEFine PROCEDURE PTInfo
1627 PMap:ch=12 :Sp$='INDEPENDANT':INK#ch,6:INK#2,6
1628 IF astro(n,7)=1:Sp$=' ALLIANCE ':INK#ch,5:INK#2,5
1629 IF astro(n,7)=2:Sp$=' REPUBLIC ':INK#ch,3:INK#2,3
1630 bx=astro(n,8):by=astro(n,9):Pop$=astro(n,12)
1631 CLS#14:PRINT#14,'GALAXY GRID '\x:bx;' y:by,' '
1632 CLS#15:PRINT#15,' STAR SYSTEM'\FILL$(' ',12-LEN(astro$(n)))&astro$(n)
1633 CLS#ch:QBold ch,7,-2,0,Sp$:CIRCLE#2,159,39,4:PTMoral
1634 IF '.' INSTR Pop$>=0:Pop$=Pop$&'0'
1635 IF '.' INSTR Pop$+1>0:Pop$=Pop$('1 TO '.' INSTR Pop$+1)
1636 PRINT#ch,FILL$(' ',6-LEN(Pop$))&Pop$&'b:Popu'
1637 PRINT#ch,FILL$(' ',6-LEN(pmv))&pmv&' :Moral'
1638 PRINT#ch,FILL$(' ',6-LEN(astro(n,1)))&astro(n,1)&' :Food'
1639 PRINT#ch,FILL$(' ',6-LEN(astro(n,2)))&astro(n,2)&' :Ores'
1640 PRINT#ch,FILL$(' ',6-LEN(astro(n,3)))&astro(n,3)&' :Tech'
1641 PRINT#ch,FILL$(' ',6-LEN(astro(n,4)))&astro(n,4)&' :Trade'
1642 PRINT#ch,FILL$(' ',6-LEN(astro(n,5)))&astro(n,5)&' :WMD'
1643 PRINT#ch,FILL$(' ',6-LEN(astro(n,6)))&astro(n,6)&' :Fuel':PMap
1644 END DEFINE

```



1646 REMark Opponent & Simulation Mode

1648 DEFine PROCedure GStarSel

```

1649 REPEAT WMD_lp
1650   sn=RND(1 TO 70):IF WMD>11 AND WMD<50:EXIT WMD_lp
1651   IF WMD>49 AND astro(sn,7)<>Gp:EXIT WMD_lp
1652   IF WMD<12 AND astro(sn,7) = Gp:EXIT WMD_lp
1653 END REPEAT WMD_lp
1654 n=sn:SSInfo:PTInfo:HJump:StarView:EOrbit
1655 IF astro(n,7)=Gp:GSTrans:ELSE GSAtt
1656 END DEFine

```

Note: Random selection of Star System
Go with any Star System
Independent or Enemy Stars (for Attack)
Annexed Stars (restock WMDs)

1658 DEFine PROCedure GSAtt

```

1659 Mes$=' Attack Mode ':SSInfo
1660 REPEAT Att_lp
1661   IF side(Gp,5)<astro(n,5)*3 OR side(Gp,5)<6:EXIT Att_lp
1662   IF astro(n,5)=0
1663     Victory:PAUSE 50:astro(n,7)=Gp:LMap:GSTrans:EXIT Att_lp
1664   END IF
1665   RollDice:SSInfo:PAUSE 50
1666 END REPEAT Att_lp
1667 END DEFine

```

Attack Mode

1669 DEFine PROCedure GSTrans

```

1670 Mes$=' ↑ ↓ ← → Transfer → ':SSInfo
1671 FOR i=1 TO 4:min(i)=1+side(Gp,5) DIV 20
1672   max(1)=20:max(2)=8:max(3)=12:max(4)=16
1673   FOR t=1 TO 4
1674     REPEAT SP_lp
1675       PAUSE 5:IF side(Gp,t)=min(t) OR astro(n,t)=max(t):EXIT SP_lp
1676       side(Gp,t)=side(Gp,t)-1:astro(n,t)=astro(n,t)+1:SSInfo
1677     END REPEAT SP_lp
1678   END FOR t
1679   ack=0:vck=1:SSInfo:PAUSE 30:Harvest:vck=0:SSInfo:PAUSE 30
1680   min(1)=1+INT(astro(n,12)):min(5)=3:min(6)=0:FOR i=2 TO 4:min(i)=1
1681   max(1)=20:max(2)=8:max(3)=12:max(4)=16:max(5)=99:max(6)=10
1682   FOR t=1 TO 6
1683     REPEAT PS_lp
1684       PAUSE 5:IF astro(n,t)=min(t) OR side(Gp,t)=max(t):EXIT PS_lp
1685       side(Gp,t)=side(Gp,t)+1:astro(n,t)=astro(n,t)-1:SSInfo
1686     END REPEAT PS_lp
1687   END FOR t
1688   IF side(Gp,5)>50 AND astro(n,5)<10
1689     astro(n,1)=astro(n,5)+5:side(Gp,5)=side(Gp,5)-5
1690   END IF
1691   SSInfo:PAUSE 30
1692 END DEFine

```

Transfer Mode

Star Ship to Planet Transfers

Planet to Star Ship Transfers

1694 DEFine PROCedure GSswap

```

1695 SSInfo:LOrbit:ws=1:PMap:bx=-9:by=-9:n=RND(20 TO 50)
1696 BLOCK#2,60,10,348,6,0:DPS 11:DiceCLS:Mes$=""
1697 IF APs=0 OR RPs=0:GEnd
1698 END DEFine

```

Move to next Turn

If Planet count zero Game End

1700 REMark Part 3 End Game

1702 DEFine PROCEDURE GEnd

```
1703 dx=60:dy=80:ws=1:DeathStar:wx=156:wy=28:CSIZE 2,1
1704 IF RPs=0
1705   BEEP 2000,20,30,0,0,0:INK 2:DSDefeat:PAUSE 10:ASS 1,7,50,20
1706   OVER 1:FOR i=1 TO 2:CURSOR 60+i,40:PRINT 'ALLIANCE':OVER 0
1707 END IF
1708 IF APs=0
1709   BEEP 2000,20,30,0,0,0:INK 2:x=56:y=80:DSAttack:PAUSE 10:RSS 1,7,50,20
1710   OVER 1:FOR i=1 TO 2:CURSOR 84+i,40:PRINT 'REPUBLIC':OVER 0
1711 END IF
1712 CSIZE 2,0:INK 7:OVER 1:FOR i=1 TO 2:CURSOR 102+i,60:PRINT 'Victory'
1713 OVER 0:CSIZE 0,0:INK 4:LName:nck=1:GMenu
1714 END DEFine
```

1716 DEFine PROCEDURE DeathStar

```
1717 FOR a=1 TO 16
1718   INK astro(n,10):wx=80+a*4:wy=-46+a*4
1719   FILL 1:CIRCLE wx,wy,50-a*2:FILL 0
1720   INK 248:FILL 1:CIRCLE 20+a*2,90-a,a/3:FILL 0:PAUSE 4
1721   INK 0:FILL 1:CIRCLE 20+a*2,90-a,a/3:FILL 0
1722   FILL 1:CIRCLE wx,wy,50-a*2:FILL 0
1723 END FOR a
1724 INK astro(n,10):FILL 1:CIRCLE wx,wy,14:FILL 0
1725 INK 248:FILL 1:CIRCLE 56,80,9:FILL 0
1726 INK 7:CIRCLE 56,80,9:INK 0:ARC 44,80 TO 68,80,PI/3
1727 INK 7:FILL 1:CIRCLE 58,84,4:FILL 0
1728 END DEFine
```

1730 DEFine PROCEDURE DSAttack

```
1731 FOR a=1 TO 6
1732   INK 2:FILL 1:LINE dx+2,dy+4 TO wx-16,wy-6 TO wx-15,wy-4 TO dx+2,dy+4
1733   FILL 0:PAUSE 3:INK 0:FILL 1
1734   LINE dx+2,dy+4 TO wx-16,wy-6 TO wx-15,wy-4 TO dx+2,dy+4:FILL 0:PAUSE 2
1735 END FOR a
1736 INK 7:FILL 1:CIRCLE dx-1,dy+4,4:FILL 0:BEEP 20000,20,10,2,3,4,5,0
1737 INK 0:FILL 1:CIRCLE wx-16,wy-6,16:FILL 0
1738 INK 7:FOR b=7 TO 10:CIRCLE wx-b,wy-12,b*4,.2,PI/2:PAUSE 5
1739 INK astro(n,10)
1740 FOR c=1 TO 12:FILL 1:CIRCLE wx-40+c*4,wy+RND(-12 TO 4),RND(2 TO 3):FILL 0
1741 END DEFine
```

1743 DEFine PROCEDURE DSDefeat

```
1744 FOR i=1 TO 18
1745   x1=56+RND(-3 TO 3)*5:y1=80+RND(-3 TO 3)*4:LINE dx,dy+2 TO x1,y1:PAUSE 1
1746 END FOR i
1747 INK 7:dx=dx-4:BEEP 20000,12,4,8,0,0,0:PAUSE 10
1748 FOR i=2 TO 4
1749   x1=dx-i*4:x2=dx-i*7:x3=dx+i*7:x4=dx+i*4:y1=dy:y2=dy+3
1750   ARC x1,y2 TO x2,y1,PI/4 TO x3,y1,PI/4 TO x4,y2,PI/4:PAUSE 5
1751 END FOR i
1752 END DEFine
```

1754 REMark Part 4 League Table

1756 DEFine PROCedure LTDefault

1757 REMark Score League Table

1758 name\$(1)="SPlay 'score(1,1)=5072:score(1,2)=1024:score(1,3)=58

1759 name\$(2)="CPlay 'score(2,1)=3765:score(2,2)=1185:score(2,3)=62

1760 name\$(3)="QBITS 'score(3,1)=3452:score(3,2)=1296:score(3,3)=66

1761 END DEFINE

League Table				
Skill	Time h:m:s	WMD	Stars	Gamer
Marshal	01:24:32	1024	58	SPlay
Admiral	01:02:45	1185	62	CPlay
Captain	00:57:32	1296	66	QBITS

1763 DEFine PROCedure LTRreset

1764 REMark Resets League Table

1765 name\$(1)=" 'score(1,1)=0:score(1,2)=0:score(1,3)=0

1766 name\$(2)=" 'score(2,1)=0:score(2,2)=0:score(2,3)=0

1767 name\$(3)=" 'score(3,1)=0:score(3,2)=0:score(3,3)=0

1768 END DEFINE

League Table				
Skill	Time h:m:s	WMD	Stars	Gamer
Marshal	00:00:00	0	0	
Admiral	00:00:00	0	0	
Captain	00:00:00	0	0	

1770 DEFine PROCedure LScore

1771 ch=13:GFile\$='QBAD2375DLT'

1772 IF LTck=0:FCHECK:IF pck=0:LTDefault:LSave:ELSE LLoad

1773 OVER#ch,1:CSIZE#ch,2,0:INK#ch,7:CLS#ch

1774 FOR i=1 TO 2:CURSOR#ch,60+i,2:PRINT#ch,'League Table'

1775 OVER#ch,0:CSIZE#ch,0,0:INK#ch,5

1776 CURSOR#ch,6,16:PRINT#ch,'Skill Time h:m:s WMD Stars Gamer'

1777 CURSOR#ch,6,28:PRINT#ch,'Marshal\' Admiral\' Captain'

1778 FOR a=1 TO 3

1779 HST\$=DATE\$(score(a,1)):HS2=score(a,2):HS3=score(a,3):INK#ch,3

1780 CURSOR#ch, 60,18+a*10:PRINT#ch,HST\$(13 TO 20)

1781 CURSOR#ch,110,18+a*10:PRINT#ch,FILL\$(' ',6-LEN(HS2))&HS2

1782 CURSOR#ch,166,18+a*10:PRINT#ch,FILL\$(' ',3-LEN(HS3))&HS3:INK#ch,6

1783 CURSOR#ch,192,18+a*10:PRINT#ch,' 'name\$(a)

1784 END FOR a

1785 END DEFINE

1787 DEFine PROCedure LName

1788 GTS=DATE-Gclk+GTS:Gclk\$=DATE\$(GTS):IF LTck=0:LScore

1789 IF score(GSk,1)>0 AND score(GSk,1)<GTS:RETURN

1790 IF Gp1=1:score(GSk,2)=AWMD+side(1,5):score(GSk,3)=APs

1791 IF Gp1=2:score(GSk,2)=RWMD+side(2,5):score(GSk,3)=RPs

1792 score(GSk,1)=GTS:BLOCK#20,50,10,390,230,0:LTck=1:LScore

1793 OPEN_NEW#99,con_60x10a10x10_10

1794 WINDOW#99,60,10,320+gx,177+gy+GSk*10:PAPER#99,0:CLS#99

1795 INK#99,6:INPUT#99,name\$(GSk):CLOSE#99:LSave

1796 END DEFINE

League Table				
Skill	Time h:m:s	WMD	Stars	Gamer
Marshal	00:00:00	0	0	
Admiral	00:00:00	0	0	
Captain	01:13:22	68	10	QBITS

1798 DEFine PROCedure LSave

1799 DELETE Drv\$(dn%)&'QBAD2375DLT':OPEN_NEW#99,Drv\$(dn%)&'QBAD2375DLT'

1800 FOR a=1 TO 3:PRINT#99,name\$(a)\score(a,1)\score(a,2)\score(a,3)

1801 CLOSE#99:LTck=1

1802 END DEFINE

1804 DEFine PROCedure LLoad


1805 OPEN_IN#99,Drv\$(dn%)&'QBAD2375DLT'

1806 FOR a=1 TO 3:INPUT#99,name\$(a)\score(a,1)\score(a,2)\score(a,3)

1807 CLOSE#99:LTck=1

1808 END DEFINE

1810 REMark File Load & Save



1812 DEFINE PROCEDURE GLoad

1813 ch=13:SelPath:IF pck=0:BLOCK#ch,250,10,0,50,0:RETURN :ELSE FCheck

1814 IF pck=0 OR eck=1

1815 CURSOR#ch,24,50:PRINT#ch,' File NOT Found':CLS#ch,4

1816 PAUSE 30:CURSOR#ch,0,50:CLS#ch,4:eck=0:RETURN

1817 END IF

1818 OPEN _IN#99,Drv\$(dn%)&Gf\$&df%:pck=0

1819 CURSOR#ch,24,50:PRINT#ch,'Loading':CLS#ch,4:OVER#ch,1

1820 FOR a=1 TO 70

1821 CURSOR#ch,70+a*2,50:PRINT#ch,':':PAUSE 1

1822 INPUT#99,astro\$(a):FOR b=1 TO 11:INPUT#99,astro(a,b):END FOR b

1823 END FOR a

1824 FOR a=1 TO 2:FOR b=1 TO 5:INPUT#99,side(a,b):END FOR b:END FOR a

1825 INPUT#99,GTS\TD2\Gp1\Gp2\AInt\RInt:CLOSE#99

1826 OVER#ch,0:AT#ch,5,0:CLS#ch,3:nck=0:GChoice

1827 END DEFINE

File NOT Found

Loading

1829 DEFINE PROCEDURE GSave

1830 ch=13:SelPath:IF pck=0:BLOCK#ch,250,10,0,50,0:RETURN :ELSE FCheck

1831 IF eck=1

1832 CURSOR#ch,24,50:PRINT#ch,'DEVICE ERROR':CLS#ch,4

1833 PAUSE 30:CURSOR#ch,24,50:CLS#ch,4:eck=0:RETURN

1834 END IF

1835 IF pck=1

1836 CURSOR#ch,24,50:PRINT#ch,'Overwrite Y/N':CLS#ch,4:PAUSE

1837 IF KEYROW(5)<>64:CURSOR#ch,24,50:CLS#ch,4:RETURN

1838 END IF

1839 DELETE Drv\$(dn%)&Gf\$&nf%:pck=0

1840 CURSOR#ch,24,50:PRINT#ch,'Saving':CLS#ch,4:OVER#ch,1

1841 OPEN_NEW#99,Drv\$(dn%)&Gf\$&df%

1842 FOR a=1 TO 70

1843 CURSOR#ch,70+a*2,50:PRINT#ch,':':PAUSE 1

1844 PRINT#99,astro\$(a):FOR b=1 TO 11:PRINT#99,astro(a,b):END FOR b

1845 END FOR a

1846 FOR a=1 TO 2:FOR b=1 TO 5:PRINT#99,side(a,b):END FOR b:END FOR b

1847 PRINT#99,GTS\TD2\Gp1\Gp2\AInt\RInt:CLOSE#99

1848 OVER#ch,0:AT#ch,5,0:CLS#ch,3

1849 END DEFINE

DEVICE ERROR

Overwrite Y/N

Saving

1851 DEFINE PROCEDURE FCheck

1852 ch=13:CURSOR#ch,24,50:PRINT#ch,'Searching...':CLS#ch,4

1853 PAUSE 20:DELETE Drv\$(dn%)&FList'

1854 OPEN_NEW#99,Drv\$(dn%)&FList':DIR#99,Drv\$(dn%):CLOSE#99

1855 OPEN_IN#99,Drv\$(dn%)&FList'

1856 REPEAT Dir_lp

1857 IF EOF(#99):CLOSE#99:BLOCK#ch,250,10,0,50,0:pck=0:EXIT Dir_lp

1858 INPUT#99,Fchk\$:IF Fchk\$==GFile\$:CLOSE#99:pck=1:EXIT Dir_lp

1859 END REPEAT Dir_lp

1860 END DEFINE


Searching...

```

1862 DEFine PROCEDURE SelPath
1863 ch=13:eck=0:INK#ch,4:BLOCK#ch,12,3,224,54,4 :Gf$='AD2375Data_'
1864 CURSOR#ch, 24,50:PRINT#ch,'Select: ↑ ↓';Drv$(dn%)&Gf$&df%
1865 CURSOR#ch,190,50:PRINT#ch,' ← → ←':BLOCK#ch,2,4,214,52,4:INK#ch,7
1866 REPEAT Path_ip
1867 CURSOR#ch, 84,50:PRINT#ch,Drv$(dn%)
1868 CURSOR#ch,180,50:PRINT#ch,df% :k=CODE(INKEY$(5))
1869 SELECT ON k
1870 =192:df%=df%-1:IF df%<0:df%=9
1871 =200:df%=df%+1:IF df%>9:df%=0
1872 =208:dn%=dn%+1:IF dn%>dm%:dn%=0
1873 =216:dn%=dn%-1:IF dn%<1:dn%=dm%
1874 = 10:pck=1:GFile$=Gf$&df%:EXIT Path_ip
1875 = 32:pck=0:EXIT Path_ip
1876 END SELECT
1877 END REPEAT Path_ip
1878 END DEFINE

```

1880 REMark **Game Init Intro & Notes**



```

1882 DEFine PROCEDURE Init_Win
1883 OPEN#20,scr_:WINDOW#20,512,256,gx,gy:BORDER#20,1,3:PAPER#20,0:CLS#20
1884 WINDOW#1,496,220,8+gx,10+gy:PAPER#1,0:CLS#1:SCALE#1,120,0,0
1885 GTitle 2,138,8,'QBITS GALAXY AD2375':GIntro
1886 WINDOW#2,500,226,6+gx,4+gy:PAPER#2,0:SCALE#2,100,0,0:INK#2,127:CLS#2
1887 FILL#2,1:LINE#2,30,37 TO 134,37 TO 127,31.5 TO 37,31.5 TO 30,37:FILL#2,0
1888 WINDOW#1,312,124,100+gx,24+gy:BORDER#1,4,248
1889 WINDOW#1,304,118,104+gx,26+gy:SCALE#1,100,0,0:CLS#1
1890 GTitle 2,158,0,'GALAXY AD2375':Emblem:RESTORE 1892
1891 FOR ch=3 TO 18:OPEN#ch,scr_:READ a,b,c,d:WINDOW#ch,a,b,c+gx,d+gy
1892 DATA 132, 46,104,26, 90, 56,211,26, 132,46,276, 26 :REMark win 3-5
1893 DATA 132, 40,104,62, 132, 40,276,62 :REMark win 6-7
1894 DATA 132, 50,104,94, 90, 62,211,82, 132,50,276, 94 :REMark win 8-10
1895 DATA 90,100, 8,28, 90,100,414,28, 274,66,120,158 :REMark win 11-13
1896 DATA 78, 22, 14, 4, 78, 22,420, 4, 198,12,154,146 :REMark win 14-16
1897 DATA 100, 62,18,158, 100, 62, 396,158 :REMark win 17-18
1898 INK#14,7:INK#15,7:INK#16,7 :SCALE#17,70,0,0:SCALE#18,70,0,0
1899 WINDOW#0,500,30,6+gx,224+gy:Init Stats:DPS 11:DPS 12
1900 sn=20:astro$(sn)='Sol ':astro$(sn,10)=228:astro$(sn,11)=2
1901 BORDER#13,1,248:BORDER#16,1,6:StarView:StarFleet
1902 END DEFINE

```

```

1904 DEFine PROCEDURE GTitle(ch,x,y,str$)
1905 OVER#ch,1:CSIZE#ch,2,1
1906 INK#ch,2:FOR a=0 TO 1:CURSOR#ch,x+a,y :PRINT#ch,str$
1907 INK#ch,6:FOR a=2 TO 3:CURSOR#ch,x+a,y+1:PRINT#ch,str$
1908 OVER#ch,0:CSIZE#ch,0,0
1909 END DEFINE

```

1911 DEFine PROCEDURE Emblem

```
1912 x=82:y=96:INK#2,6:CIRCLE#2,x,y,4:3.2
1913 FILL#2,1:LINE#2,x-2.5,y TO x-1.2,y+2 TO x+1.2,y+2 TO x+2.5,y
1914 LINE#2 TO x+1.2,y-1.9 TO x-1.2,y-1.9 TO x-2.5,y:FILL#2,0
1915 INK#2,0:FILL#2,1:CIRCLE#2,x,y,1:FILL#2,0:LINE#2,x-2.5,y TO x+2.5,y
1916 LINE#2,x+1.8,y+2 TO x-2,y-2.2:LINE#2,x-1.8,y+2 TO x+2,y-2.4
1917 END DEFine
```



1919 DEFine PROCEDURE GIcon

```
1920 ch=20:x=74:y=7:IF SPck=0:INK#ch,6:ELSE INK#ch,2
1921 LINE#ch,x-35,y TO x,y+4 TO x+35,y TO x,y-4 TO x-35,y
1922 FILL#ch,1:CIRCLE#ch,x,y,3,.3,PI/2:FILL#ch,0
1923 END DEFine
```



1925 DEFine PROCEDURE Init_Stats

```
1926 CLS#17:INK#17,5:CUSOR#17,0,2:CLS#18:INK#18,3:CUSOR#18,0,2
1927 PRINT#17,'WMD' :PRINT#18,' WMD'
1928 QBold 17,7,25,2,'ALLIANCE' :QBold 18,7,1,2,'REPUBLIC'
1929 PRINT#17,' :Land Based' :PRINT#18,' Land Based:'
1930 PRINT#17,' : ' :PRINT#18,' : '
1931 PRINT#17,' : ' :PRINT#18,' : '
1932 PRINT#17,'Fuel:' :PRINT#18,' :Fuel'
1933 PRINT#17,'Star Systems:' :PRINT#18,' :Star Systems'
1934 Alliance:Republic:SFuel:IF Gp1=1:ch=17:x=25:ELSE ch=18:x=5
1935 LINE#ch,x,69 TO x+50,69 TO x+50,57 TO x,57 TO x,69
1936 ASS 17,7,45,36:AF2 17,22,28:INK#18,7:RSS 18,7,40,34:RF2 18,62,28
1937 LINE#2,2,0 TO 2,29:ARC#2 TO 4,33,-PI/2:LINE#2 TO 18,33 TO 22,36
1938 LINE#2,162,0 TO 162,29:ARC#2 TO 160,33,PI/2:LINE#2 TO 146,33 TO 142,36
1939 END DEFine
```



1941 DEFine PROCEDURE Alliance

```
1942 ch=2:x=25.1:y=38:INK#ch,7
1943 FILL#ch,1:ARC#ch,x-2.8,y TO x+2.8,y,-PI/1.2:FILL#ch,0
1944 LINE#ch,x+2.8,y TO x+2.5,y-3:LINE#ch,x-2.8,y TO x-2.5,y-3
1945 INK#ch,0:LINE#ch,x-4,y+.8 TO x-4,y+2.8:LINE#ch,x+4,y+.8 TO x+4,y+2.8
1946 INK#ch,7:LINE#ch,x-2,y+.8 TO x-2,y+2.8:LINE#ch,x+2,y+.8 TO x+2,y+2.8
1947 ARC#ch,x-2,y-.5 TO x-2,y-.5,PI:LINE#ch TO x-2,y-.5
1948 ARC#ch,x+2,y-.5 TO x+2,y-.5,PI:LINE#ch TO x+2,y-.5
1949 CIRCLE#ch,x,y-3.2,1.2,.6,PI:LINE#ch,x-2.5,y-3 TO x,y-4 TO x+2.5,y-3
1950 END DEFine
```



1952 DEFine PROCEDURE Republic

```
1953 ch=2:x=138.6:y=38:INK#ch,7
1954 FILL#ch,1:ARC#ch,x-2.5,y TO x+2.5,y,-PI/1.5:LINE#ch TO x+3,y-3.6
1955 LINE#ch TO x+1.4,y-5 TO x-1.4,y-5 TO x-3,y-3.6 TO x-2.5,y:FILL#ch,0
1956 INK#ch,0:LINE#ch,x-2,y-3.8 TO x,y-3 TO x+2,y-3.8
1957 FILL#ch,1:ARC#ch,x-2,y-.5 TO x-2,y-.5,PI:LINE#ch TO x-2,y-.5:FILL#ch,0
1958 FILL#ch,1:ARC#ch,x+2,y-.5 TO x+2,y-.5,PI:LINE#ch TO x-2,y-.5:FILL#ch,0
1959 FILL#ch,1:LINE#ch,x-2,y-5 TO x,y-4 TO x+2,y-5 TO x-2,y-5:FILL#ch,0:INK#2,7
1960 END DEFine
```



1962 DEFine PROCEDURE StarFleet

```
1963 ASS 1,7,44,38:AF1 1,20,46:AF2 1,44,58:AF3 1,66,46
1964 RSS 1,7,148,36:RF1 1,124,46:RF2 1,148,56:RF3 1,170,46
1965 END DEFine
```



```

1967 DATA 'Vega' , 'Bellatrix', 'Zibal' , 'Castor', 'Naos' , 'Artos' , 'Ilad'
1968 DATA 'Tabit' , 'Diphda' , 'Elnath' , 'Acrux' , 'Errai' , 'Cellus' , 'Detrus'
1969 DATA 'Polaris', 'Homan' , 'Rigel' , 'Sabik' , 'Sarin' , 'Westron', 'Zenda'
1970 DATA 'Furud' , 'Gienah' , 'Hadar' , 'Propus', 'Izar' , 'Centra' , 'Nebtron'
1971 DATA 'Enif' , 'Algol' , 'Mirzarm', 'Libeta', 'Atria' , 'Panus' , 'Ulan'
1972 DATA 'Keid' , 'Kochab' , 'Lesath' , 'Ankaa' , 'Marsic' , 'Nebus' , 'Alean'
1973 DATA 'Meissa' , 'Subra' , 'Menkib' , 'Altair' , 'Muscida' , 'Pusan' , 'Curson'
1974 DATA 'Cursa' , 'Nashira' , 'Nunki' , 'Ogma' , 'Pollux' , 'Xilon' , 'Scarb'
1975 DATA 'Sirius' , 'Merack' , 'Serena' , 'Denab' , 'Tureis' , 'Noran' , 'Theron'
1976 DATA 'Beid' , 'Wesen' , 'Yildun' , 'Caph' , 'Zooma' , 'Jarron' , 'Cetrus'

```

1978 **DEFine PROCedure Init New**

1979 ch=13:Ac=0:Rc=0:INK#ch,7:**RESTORE 1967**

1980 CURSOR#ch,18,50:CLS#ch,4:PRINT#ch,'Initilising':OVER#ch,1

1981 **FOR n=1 TO 70**

1982 **READ astro\$(n):**CURSOR#ch,92+n*2,50:PRINT#ch,'':PAUSE 1

1983 Popu=RND(4 TO 16)/RND(2 TO 8) :REMark Population 0.5<8bn

1984 astro(n,1)=INT(Popu*2.5) :REMark Food max 20

1985 astro(n,2)=INT(Popu*2) :REMark Ores max 10

1986 astro(n,3)=INT(Popu*1.5) :REMark Tec max 12

1987 astro(n,4)=INT((Popu+astro(n,2))/2) :REMark Trade max 20

1988 astro(n,5)=INT((Popu+astro(n,3))/2) :REMark WMD max 20

1989 astro(n,6)=INT(astro(n,2)/3) :REMark Fuel Cells

1990 astro(n,7)=0 :REMark Side 0=Independent 1=Alliance 2=Republic

1991 astro(n,8)=n*4+15 :REMark x Grid coordinate

1992 astro(n,9)=RND(10 TO 90) :REMark y Grid coordinate

1993 astro(n,10)=RND(24 TO 46) :REMark Planet Colour

1994 astro(n,11)=RND(1 TO 3) :REMark Solar System

1995 astro(n,12)=Popu :REMark Population max 10bn

1996 **END FOR n**

1997 n=RND(5 TO 65):astro(n,7)=1:astro(n,5)=12:astro(n,6)=10

1998 n=RND(35 TO 45):astro(n,7)=2:astro(n,5)=12:astro(n,6)=10

1999 **REPEAT Count_lp**

2000 n=RND(20 TO 50):IF Rc<9 AND astro(n,7)=0:astro(n,7)=2:Rc=Rc+1

2001 n=RND(5 TO 30):IF Ac<9 AND astro(n,7)=0:astro(n,7)=1:Ac=Ac+1

2002 n=RND(40 TO 65):IF Ac<9 AND astro(n,7)=0:astro(n,7)=1:Ac=Ac+1

2003 IF Ac=9 AND Rc=9:EXIT Count_lp

2004 **END REPEAT Count_lp**

2005 FOR n=1 TO 70 :IF astro(n,7)>0 AND astro(n,5)<3:astro(n,5)=3

2006 FOR s=1 TO 2 :side(s,1)=4:side(s,2)=2:side(s,3)=6:side(s,4)=6

2007 OVER#ch,0:Alnt=12:Rlnt=12:sch=0:**GMaplnit:LMap**:GTS=0:nck=0

2008 **END DEFine**

2010 DEFINE PROCEDURE GIntro

```

2011 FOR Stars=1 TO 250:INK RND(2 TO 7):POINT RND(0 TO 280),RND(0 TO 140)
2012 CSIZE 0,0:RESTORE 2017:ch=1:FOR a=1 TO 12:READ i,c,r,str$:MPrn :PAUSE 5
2013 ASS 1,7,40,72:RSS 1,7,160,70:PAUSE
2014 FOR a=1 TO 20:BLOCK 12+a*20,200,240-a*10,20,0:PAUSE 2
2015 END DEFINE

```



```

2017 DATA 5,110,30,'There is Disquiet in the Galaxy as the Republic'
2018 DATA 5,126,40,'First Order seeks to enforce new Taxes on'
2019 DATA 5,156,50,'Interstellar Travel and Trade.'
2020 DATA 3,158,74,'(M)arshal (G)eneral (C)aptain'
2021 DATA 6,166,86,'Choose Skill Level and Side'
2022 DATA 3,173,98,'(A)lliance or (R)epublic'
2023 DATA 5,96,122,'Select Star System and <HyperSpace Jump> to Planet'
2024 DATA 5,96,132,'Attack <?> Annex for WMD/Trade/Tech/Ores/Food/Fuel'
2025 DATA 5,116,142,'Harvest Results and Reassign to Defend your'
2026 DATA 5,132,152,'Star Sytems or use to Attack the Enemy'
2027 DATA 3,186,170,'(G)ame Notes (M)enu';7,206,192,'Press Any key'
2028 DATA 2,26,70,'(F1) Simulation Mode (F2) Graphics Check'
2029 DATA 2,26,80,'(F3) Timeslip Check - Swap Sides/Assets'
2030 DATA 2,26,90,'(F4) Enemy Encounter (F5) Tech/WMD Cheat'
2031 DATA 5,50,104,'(D)efault - Highscore - Reset(d)'

```

2033 DEFINE PROCEDURE GNotes

```

2034 ch=1:INK 6:CUSOR 76,4:PRINT 'GALAXY AD2375 Game Notes':INK 4
2035 RESTORE 2023:FOR a=1 TO 5:READ i,c,r,str$:c=c-96:r=6+a*10:MPrn
2036 RESTORE 2028:FOR a=1 TO 4:READ i,c,r,str$:MPrn
2037 INK 7:OVER 1:CUSOR 96,56:PRINT 'G M':OVER 0
2038 PAUSE 500:CLS:FOR i=1 TO 120:POINT RND(10 TO 200),RND(5 TO 95)
2039 END DEFINE

```

2041 DEFINE PROCEDURE MPPrn:INK i:CUSOR#ch,c,r:PRINT#ch,str\$:END DEFINE

```

:
2043 DEFINE PROCEDURE GameTest
2044 Pulsar:Binary:Nebula:Blackhole:TimeSlip:HJump:StarView
2045 SPck=1:EOrbit:RoIDice:SSInfo:LORbit:Encounter
2046 FOR n=1 TO 70:astro(n,5)=RND(5 TO 12):astro(n,7)=Gp1
2047 LMap:SSStats:ch=1:nck=1:SPck=0:GEnd
2048 END DEFINE

```